

1 Grundprinzip

Boolesche Operationen verknüpfen mehrere Körper. Die Körper sollten sich dabei eindeutig überlappen. Geschweifte Klammern enthalten die beteiligten Formen.

```
operation() {  
    erster_koerper();  
    zweiter_koerper();  
}
```

2 Vereinigen: union()

Alle enthaltenen Körper werden zu einer Form verbunden. Die Körper dürfen sich überlappen oder berühren.

```
union() {  
    cube([30, 20, 6], center=true);  
    translate([15, 0, 6])  
        cylinder(h=12, d=10, center=true);  
}
```

3 union() oft optional

Oberste Körper werden gemeinsam exportiert. union() verdeutlicht die Absicht und hilft beim Verschachteln.

4 Abziehen: difference()

Der erste Körper ist das Ausgangsmaterial. Alle folgenden Körper werden davon abgezogen. Die Reihenfolge ist daher entscheidend.

```
difference() {  
    // Grundkörper  
    cube([50, 30, 6], center=true);  
  
    // Werkzeugkörper: Bohrung  
    cylinder(h=8, d=8, center=true);  
}
```

5 Mehrere Ausschnitte

Nach dem Grundkörper dürfen mehrere Werkzeuge folgen. Hier entstehen zwei Bohrungen und ein Schlitz.

```
difference() {  
    cube([60, 35, 6], center=true);  
    translate([-20,0,0])  
        cylinder(h=8,d=5,center=true);  
    translate([20,0,0])  
        cylinder(h=8,d=5,center=true);  
    cube([18, 6, 8], center=true);  
}
```

6 Schnittmenge: intersection()

Nur der gemeinsame Bereich aller Körper bleibt erhalten. Nützlich zum Beschneiden oder Erzeugen besonderer Formen.

```
intersection() {  
    cube([30, 30, 30], center=true);  
    sphere(d=38);  
}  
// Ergebnis: abgerundeter Würfelbereich
```

7 Saubere Überschneidung

Werkzeugkörper etwas größer als das Material anlegen. Deckungsgleiche Flächen können Darstellungsfehler erzeugen.

```
epsilon = 0.1;  
cylinder(  
    h=staerke+2*epsilon, d=5, center=true  
);
```

Merke

Bei difference() bleibt nur vom ersten Körper Material übrig.

8 Operationen verschachteln

Operationen können innerhalb anderer Operationen stehen. Klammern und Einrückungen zeigen die Struktur. Hier wird ein Verbundkörper von einem Quader abgezogen.

```
difference() {
  cube([60,40,8], center=true);
  union() {
    cylinder(h=10,d=12,center=true);
    cube([30,5,10],center=true);
  }
}
```

9 Werkzeugkörper prüfen

Mit # wird ein abzuziehender Körper hervorgehoben. Mit % kann der Grundkörper transparent erscheinen. Die Markierungen ändern den Export nicht dauerhaft.

```
difference() {
  %grundplatte();
  #lochbild();
}
```

10 Typische Fehler

- Werkzeug schneidet Material nicht vollständig
- Grundkörper steht nicht an erster Stelle
- fehlende Klammer oder falsche Verschachtelung

11 Praxisbeispiel: Montageplatte

Vier Zylinder werden per Schleife abgezogen. Das Z-Übermaß sorgt für vollständig offene Bohrungen.

```
breite = 80;
hoehe = 50;
staerke = 4;
loch_d = 5;
rand_x = 10;
rand_y = 10;
$fn = $preview ? 24 : 64;

difference() {
  cube([breite,hoehe,staerke], center=true);

  for (x=[-1,1], y=[-1,1])
    translate([
      x*(breite/2-rand_x),
      y*(hoehe/2-rand_y), 0
    ])
    cylinder(
      h=staerke+2,
      d=loch_d, center=true
    );
}
```

12 Wann welche Operation?

union()	Körper zusammenfügen
difference()	Bohrungen und Ausschnitte
intersection()	gemeinsamen Bereich behalten
Oft werden alle drei Operationen kombiniert.	

13 Performance

Viele Rundungen und Verschachtelungen verlängern das Rendern. In der Vorschau eine kleinere Segmentzahl verwenden. Teilmodule einzeln mit ! kontrollieren.

```
$fn = $preview ? 24 : 96;
!lochbild();
```

Mini-Übung

Baue einen 60 × 40 × 8 mm Block. Ziehe mittig eine Bohrung d=12 und zwei Langlöcher ab. Runde Variante: Langlöcher mit hull() erzeugen.